
6 C O N T R O L D E L R A Z O N A M I E N TO

Hemos visto como el programador puede controlar la ejecución de un programa or-
denando sus cláusulas y sus metas. Ahora revisaremos otra herramienta de control
conocida como corte, para prevenir la reconsideración característica de Prolog (back-
tracking). Este operador extiende además la expresividad de lenguaje, permitiendo
la definición de una forma de negación, llamada negación por fallo finito (NAF),
asociada al supuesto del mundo cerrado (CWA). En lo que sigue revisaremos estos
dos conceptos en detalle.

6.1 corte
El árbol-SLD de una meta definitiva puede tener muchas ramas que conducen al
fallo de la meta y muy pocas, ó una sola rama, que conducen al éxito. Por ello, el pro-
gramador podría querer incluir información de control en sus programas, para evi-
tar que el intérprete construya ramas fallidas. Observen que esta meta-información
se basa en la semántica operacional del programa, por lo que el programador debe
saber como se construyen y se recorren los arboles-SLD. El predicado !/0 denota la
operación de corte, y puede utilizarse como una literal en las metas definitivas. Su Corte

presencia impide la construcción de ciertos sub-arboles.
Un intérprete de Prolog recorre los nodos de un árbol-SLD primero en profun-

didad. El orden de las ramas corresponde al orden textual de las cláusulas en el
programa. Cuando una hoja es alcanzada, el proceso de backtracking es ejecutado. Backtrackig

El proceso termina cuando no es posible hacer backtracking (todos los sub-arboles
de la raíz del árbol han sido visitados).

Ejemplo 6.1. Asumamos el siguiente programa que define que el padre de una persona es
su antecesor hombre:

padre(X, Y) ← progenitor(X, Y), hombre(X).

progenitor(benjamin, antonio).

progenitor(maria, antonio).

progenitor(samuel, benjamin).

progenitor(alicia, benjamin).

hombre(benjamin).

hombre(samuel).

El árbol-SLD de la meta← padre(X, antonio) se muestra en la Figura 6.1. Bajo la función
de selección implementada en Prolog, encontrará la solución X/benjamin. El intento por
encontrar otra solución con X/maria, mediante el backtracking, fallará puesto que maria
no satisface el predicado hombre/1.

Para detallar la semántica del corte, es necesario introducir algunos conceptos
auxiliares. En un árbol-SLD, cada nodo ni corresponde a una meta Gi de una
derivación-SLD y tiene un átomo seleccionado asociado αi:

G0
α0 G1 . . . Gn−1

αn−1 Gn

Asumamos que para cierto nodo nk, αk no es una sub-meta de la meta ini-
cial. Entonces αk es un átomo βi del cuerpo de una cláusula de la forma β0 ←

← padre(X, antonio)

← progenitor(X, antonio), hombre(X).

← hombre(benjamin).

�

← hombre(maria)

Figura 6.1: Arbol de derivación-SLD para la meta← padre(X, antonio)

← padre(X, antonio)

← progenitor(X, antonio), hombre(X), !.

hombre(benjamin).

←!

�

Figura 6.2: Arbol de derivación-SLD para la meta← padre(X, antonio) con las ramas fallidas
podadas.

β1, . . . , βi, . . . , βn cuya cabeza β0 unifica con la sub-meta seleccionada en algún no-
do n0<j<k, es decir un nodo entre la raíz del árbol y el nodo nk. El nodo nj se conoce
como el origen, o padre, de αk y se denota como origen(αk). Nodo origen

El predicado de corte ! se procesa como un átomo ordinario situado en el cuerpo
de una cláusula. Sin embargo, cuando el corte es seleccionado para computar la
resolución, éste tiene éxito inmediatamente (con la substitución vacía ε como resul-
tado). El nodo donde ! fue seleccionado es llamado el nodo de corte. Un nodo de Nodo de corte

corte puede ser visitado nuevamente durante el backtracking. En este caso, el cur-
so normal del recorrido del árbol es alterado (por definición el recorrido continua
en el nodo superior a origen(!). Si el corte ocurre en la meta inicial, la ejecución
simplemente termina.

Ejemplo 6.2. La formulación del problema padre, nos dice que a lo más existe una solu-
ción para nuestra meta. cuando la solución se encuentra, la búsqueda puede detenerse pues
ninguna persona tiene más de un padre. Para forzar esta situación, el predicado de corte se
agrega al final de padre/2:

padre(X, Y)← progenitor(X, Y), hombre(X), !.

Observen que el programa modificado en el ejemplo anterior sólo puede compu-
tar un elemento de la relación padre/2. El corte detendrá la búsqueda después de
encontrar la primer respuesta para la meta ← padre(X, Y). El origen del corte es
la raíz del árbol, por lo que la búsqueda termina después de hacer backtracking al
nodo de corte. La otra rama del árbol no es recorrida. El árbol-SLD del programa
que incluye el corte se muestra en la figura 6.2.

Observen que la versión modificada con el corte, no puede usarse para computar
más de un elemento de la relación “es padre de”. El corte detendrá la búsqueda
después de encontrar la primer respuesta a la meta definitiva.

A partir de la definición del corte, se sigue que los efectos del operador son:

1. Divide el cuerpo de la meta en dos partes –Después de éxito de !/0 , no
es posible hacer backtracking hacía las literales a la izquierda del corte. Sin
embargo, a la derecha del corte todo funciona de manera usual.

2. Poda las ramas sin explorar directamente bajo origen(!). En otras palabras, no
habrá más intentos de unificar la sub-meta seleccionada de origen(!) con el
resto de las cláusulas del programa.

El corte es controvertido. La intención al introducir el corte, es poder controlar la
ejecución de los programas, sin cambiar su significado lógico. Por tanto, la lectura
lógica del corte es true. Operacionalmente, si el corte remueve sólo ramas fallidas
del árbol-SLD, no tiene influencia en el significado lógico de un programa. Pero
el corte puede remover también ramas exitosas del árbol-SLD, atentando contra la
completitud de los programas definitivos, o la correctez de los programas generales.

Ejemplo 6.3. Es bien sabido que los padres de un recién nacido están orgullosos. La propo-
sición puede representarse con la siguiente cláusula definitiva:

orgulloso(X)← padre(X, Y), recienNacido(Y).

consideren las siguiente cláusulas adicionales:

padre(X, Y) ← progenitor(X, Y), hombre(X).

progenitor(juan, maria).

progenitor(juan, cristina).

hombre(juan).

recienNacido(cristina).

La respuesta a la meta← orgulloso(juan) es true, puesto que como describimos, juan es
padre de cristina, que es un recién nacido. Ahora, si remplazamos la primera cláusula, con
su versión que utiliza corte:

padre(X, Y)← progenitor(X, Y), hombre(X), !.

Y preguntamos nuevamente a Prolog, si

← orgulloso(juan).

la respuesta será false. Esto se debe a que la primer hija de juan en el programa es maria.
Una vez que esta respuesta se ha encontrado, no habrá más intentos de satisfacer la meta en
origen(!). No se considerarán más hijos de juan en la solución computada.

El programa del ejemplo anterior se ha vuelto incompleto, algunas respuestas co-
rrectas no pueden ser computadas. Más grave aún es el caso de las metas generales,
donde se puede llegar a resultados incorrectos, por ejemplo, ← ¬orgulloso(juan)
tendría éxito en la versión de nuestro programa que utiliza corte.

Hasta ahora hemos distinguido dos usos del corte: eliminar ramas fallidas en el
árbol-SLD; y podar ramas exitosas. Eliminar ramas fallidas se considera una prác-
tica sin riesgo, porque no altera las respuestas producidas durante la ejecución de
un programa. Tales cortes se conocen como cortes verdes. Sin embargo, este uso Corte verde

del operador corte, esta ligado al uso particular de un programa. Como se ilustra
en los ejemplos anteriores, para algunas metas, el operador solo eliminará ramas
fallidas; pero para otras podará ramas exitosas. Cortar ramas exitosas se considera
una práctica de riesgo. Por eso, tales cortes se conocen como cortes rojos. Corte rojo

Ejemplo 6.4. Consideremos un ejemplo de corte verde. Si en el ejemplo anterior maria
es una recién nacida, agregaríamos la cláusula recienNacido(maria) a nuestro programa.
Entonces la meta← orgulloso(X) nos diría que X/juan está orgulloso. Esto es, juan tiene
una doble razón para estar orgulloso. Pero a nosotros nos basta con saber sólo una vez, que
orgulloso está juan. Para evitar que Prolog nos de la respuesta dos veces, definiríamos:

orgulloso(X)← padre(X, Y), recienNacido(Y), !.

Ejemplo 6.5. Ahora consideren un ejemplo de corte rojo:

min(X, Y, X) ← X < Y, !.

min(X, Y, Y).

Aparentemente nuestro programa es correcto. De hecho, el programa respondería de ma-
nera correcta a metas como ← min(2, 3, X) respondiendo que “Si” para X/2; y para
← min(3, 2, X) respondería que “Si” para X/2. Sin embargo el programa no es correc-
to. Consideren la meta ← min(2, 3, 3) y verán que Prolog respondería “Si”. La razón de
esto es que la segunda cláusula dice: el menor de X e Y es siempre Y. El corte está eliminan-
do algunas ramas fallidas, que serían útiles en la definición de min. La definición correcta,
usando corte, sería:

min(X, Y, X) ← X < Y, !.

min(X, Y, Y) ← X ≥ Y.

6.1.1 Caso de estudio

Prolog reconsidera automáticamente si esto es necesario para satisfacer una meta.
Esto es útil en el sentido que le evita al programador contender explícitamente
con la reconsideración, pero si no tenemos ninguna forma de control sobre el back-
tracking, éste puede volverse la causa de programas ineficientes. Comencemos por
estudiar un programa muy simple que involucra reconsideración, para identificar
los puntos donde el backtracking es inútil y condice a la ineficiencia.

Consideren una regulación sobre el nivel de alerta de la contaminación en una
ciudad. La Figura 6.3 muestra la relación entre la concentración de los contaminan-
tes en el aire X y el estado de la alarma Y a la manera de un semáforo. La relación
entre X e Y se puede establecer mediante tres reglas:

regla 1. Si X < 3 entonces Y = verde

regla 2. Si 3 ≤ X y X < 6 entonces Y = amarilla

regla 3. Si 6 ≤ X entonces Y = roja

0 3 6

verde amarilla roja

Estado de la alarma Y

Concentración X

Figura 6.3: Estado de la alarma ambiental en función del nivel de contaminación.

Lo cual puede escribirse en Prolog de la siguiente manera:

1 % Concentración X de contaminantes en el aire

2 % f(X,Y), Y estado de la alarma

3

4 f(X,verde) :- X<3. % Regla 1

5 f(X,amarilla) :- 3=< X, X<6. % Regla 2

6 f(X,roja) :- 6 =< X. % Regla 3

El programa asume que X será instanciada antes de la llamada a f /2, tal y como
lo requieren los operadores de comparación utilizados en su definición. A continua-
ción ejecutaremos dos experimentos, donde la ineficiencia de nuestra implementa-
ción se hará evidente; pero evitable con la ayuda del operador de corte.

Experimento 1

Supongamos que el nivel de contaminación en la ciudad es X = 2. Asumamos que
el usuario del sistema no está muy al tanto de las normativas correspondientes y
se pregunta si tal nivel de contaminación es inseguro y debiese generar una alarma
amarilla. Para ello, podría preguntarle a Prolog si:

1 ?- f(2,Y), Y=amarilla.

¿Cómo computa Prolog su respuesta? Al resolver la primer meta f(2,Y), Y uni-
ficará con verde; de forma que la segunda meta toma la forma verde=amarilla, lo
cual es falso y en consecuencia la lista completa de metas falla. Sin embargo, antes
de rendirse, Prolog evaluará vía su reconsideración dos alternativas que resultan
inútiles. Invoquen la meta bajo el modo de traza de Prolog para observar esto en
detalle:

1 [trace] ?- f(2,Y), Y=amarilla.

2 Call: (9) f(2, _5824) ?

3 Call: (10) 2<3 ?

4 Exit: (10) 2<3 ?

5 Exit: (9) f(2, verde) ?

6 Call: (9) verde=amarilla ?

7 Fail: (9) verde=amarilla ?

8 Redo: (9) f(2, _5824) ?

9 Call: (10) 3=<2 ?

10 Fail: (10) 3=<2 ?

11 Redo: (9) f(2, _5824) ?

12 Call: (10) 6=<2 ?

13 Fail: (10) 6=<2 ?

14 Fail: (9) f(2, _5824) ?

15 false.

Las tres reglas de nuestro programa son mutuamente excluyentes, de forma que
solo una de ellas tendrá éxito. Por ello, sabemos que tan pronto como una de ellas
tiene éxito, no tiene caso intentar probar las otras, puesto que fallarán. Pero Prolog
no sabe esto y para ello será necesario indicarle que no reconsidere en ciertos sitios
del programa, en este caso, al final de las primeras dos reglas:

1 % Concentración X de contaminantes en el aire

2 % f(X,Y), Y estado de la alarma

3

4 f(X,verde) :- X<3. % Regla 1

5 f(X,amarilla) :- 3=< X, X<6. % Regla 2

6 f(X,roja) :- 6 =< X. % Regla 3

cuyo efecto en la traza del programa puede verse a continuación:

1 [trace] ?- f(2,Y), Y=amarilla.

2 Call: (9) f(2, _9422) ?

3 Call: (10) 2<3 ?

4 Exit: (10) 2<3 ?

5 Exit: (9) f(2, verde) ?

6 Call: (9) verde=amarilla ?

7 Fail: (9) verde=amarilla ?

8 false.

Observen que el comportamiento de ambos programs, con y sin corte, es el mis-
mo. La única diferencia es que el primero tardará un poco más en dar su respuesta.
En estos casos se dice que el operador corte solo cambia el significado procedural
del programa. Como veremos más adelante, éste no es siempre el caso.

Experimento 2

Ejecutemos otro experimento con la versión con cortes de nuestro programa. Pre-
guntémosle la siguiente meta:

1 ?- f(7,Y).

2 Y=roja

Revisemos la traza del programa:

1 [trace] ?- f(7,Y).

2 Call: (8) f(7, _10108) ?

3 Call: (9) 7<3 ?

4 Fail: (9) 7<3 ?

5 Redo: (8) f(7, _10108) ?

6 Call: (9) 3=<7 ?

7 Exit: (9) 3=<7 ?

8 Call: (9) 7<6 ?

9 Fail: (9) 7<6 ?

10 Redo: (8) f(7, _10108) ?

11 Call: (9) 6=<7 ?

12 Exit: (9) 6=<7 ?

13 Exit: (8) f(7, roja) ?

14 Y = roja.

Las tres reglas son intentadas antes de poder computar la respuesta. Esto revela otra
fuente de ineficiencia: algunos de los tests que se llevan a cabo son redundantes.
Una tercer versión de nuestro programa sería como sigue:

1 % Tercer versión

2

3 f(X,verde) :- X<3, !.

4 f(X,amarilla) :- X<6, !.

5 f(_,roja).

Observen la traza de la meta:

1 [trace] ?- f(7,Y).

2 Call: (8) f(7, _13718) ?

3 Call: (9) 7<3 ?

4 Fail: (9) 7<3 ?

5 Redo: (8) f(7, _13718) ?

6 Call: (9) 7<6 ?

7 Fail: (9) 7<6 ?

8 Redo: (8) f(7, _13718) ?

9 Exit: (8) f(7, roja) ?

10 Y = roja.

La respuesta es la misma, pero esta versión del programa es más eficiente que las
dos anteriores. Pero, ¿Qué sucede si removemos los operadores de corte? Entonces
la salida de la meta sería:

1 Y=verde;

2 Y=amarilla;

3 Y=roja;

4 false

Esto sugiere que a diferencia de nuestro segundo programa, los cortes aquí ¡cam-
bian el resultado del programa! ¿A qué se debe la respuesta de nuestro tercer pro-
grama a las siguientes metas?

1 ?- f(2,amarilla).

2 true

3 ?- f(2,Y), Y=amarilla.

4 false

La traza de la primer meta explica lo sucedido:

1 [trace] ?- f(2,amarilla).

2 Call: (8) f(2, amarilla) ?

3 Call: (9) 2<6 ?

4 Exit: (9) 2<6 ?

5 Exit: (8) f(2, amarilla) ?

6 true.

Si no tenemos cuidado con el corte, podemos cambiar la semántica de nuestros
programas inadvertidamente.

Una descripción más precisa del funcionamiento del corte es como sigue: Llama-
remos meta padre a la meta que unifica con la cabeza de la cláusula que contiene Meta padre

al operador corte. Cuando el corte es encontrado como una meta, tiene éxito in-
mediatamente, pero compromete al sistema con todas las elecciones hechas entre
el momento en que la meta padre fue invocada y el momento en que el corte fue
encontrado. Todas las alternativas entre el nodo padre y el corte son descartadas.

Ejemplo 6.6. Consideren la siguiente cláusula:

H : −B1, B2, . . . , Bm, !, Bm+2, Bn

Asumamos que H unifica con la meta G, por lo que G es su meta padre. Al alcanzar el corte,
la solución para las metas B1, . . . Bm se fija, todas sus alternativas son descartadas. No así
las metas Bm+2, . . . , Bn.

6.1.2 Otros ejemplos

Implementaremos una función miembro determinista, e.g., solo computa una res-
puesta:

1 % single member

2

3 smember(X,[X|L]) :- !.

4 smember(X,[_,L]) :- smember(X,L).

De esta forma, su comportamiento es como sigue:

1 ?- smember(X,[1,2,3]).

2 X=1;

3 false

A veces queremos agregar un elemento a una lista, solo en el caso de que éste no
sea ya parte de ella. La siguiente función hace este trabajo:

1 % agregar sin duplicar

2

3 add(X,L,L) :- smember(X,L), !.

4 add(X,L,[X|L]).

Al igual que en el caso de max/3 se asume que el tercer argumento de add/3 no
está instanciado en las llamadas. En caso contrario, su comportamiento es inespera-
do, por ejemplo:

1 ?- add(a,[a],[a,a]).

2 true

Nuestro último ejemplo tiene que ver con clasificar objetos en categorías que
cumplen ciertas condiciones y son excluyentes entre si. Por ejemplo, asumamos las
siguientes relaciones entre equipos de futbol:

1 % clasificando en categorias

2

3 derrota(barcelona, real_madrid).

4 derrota(roma, barcelona).

5 derrota(cruz_azul, real_madrid).

La idea es clasificar a los equipos mediante una relación clase/2 siguiendo los
siguientes criterios:

ganador. Es un equipo que gana todos sus encuentros.

luchador. Es un equipo que gana algunos de sus encuentros, pero pierde otros.

deportista. Un equipo que pierde todos sus encuentros.

La definición de clase/2 en Prolog es como sigue:

1 % luchador es quien gana y pierde algunos partidos

2 % ganador es quien gana siempre

3 % deportista es quien pierde siempre

4

5 clase(X,luchador) :-

6 derrota(X,_),

7 derrota(_,X), !.

8

9 clase(X, ganador) :-

10 derrota(X,_), !.

11

12 clase(X, deportista) :-

13 derrota(_, X).

La consulta al programa es como sigue:

1 ?- clase(barcelona,X).

2 X = luchador;

3 false

4 ?- clase(barcelona, deportista).

5 false

6.2 negación por fallo finito
¿Cómo podemos expresar la siguiente proposición en Prolog: A Adriana le gustan
todos los animales, excepto las serpientes? La primer parte de la proposición es
sencilla:

1 le_gusta(adriana,X) :-

2 animal(X).

Pero la segunda parte es más complicada. Deberíamos expresar que si X es una
serpiente, entonces es falso que X le gusta a adriana; y que en cualquier otro caso
X le gusta a adriana. La traducción a Prolog es la siguiente:

1

2 % A Adriana le gustan todos los animales, excepto las

3 % serpientes

4

5 le_gusta(adriana,X) :-

6 serpiente(X), !, fail.

7

8 le_gusta(adriana,X) :-

9 animal(X).

De hecho, las dos cláusulas puede escribirse juntas de manera más compacta:

1 le_gusta(adriana,X) :-

2 serpiente(X), !, fail

3 ;

4 animal(X).

Podemos usar la misma idea, para escribir un predicado di f erente/2:

1 diferente(X,Y) :-

2 X=Y, !, fail

3 ;

4 true.

Estos ejemplos sugieren que sería conveniente tener un predicado not/1 con la
siguiente definición:



6.2 negación por fallo finito 101

1 not(P) :-

2 P,!,fail

3 ;

4 true.

Esta definición descansa enteramente en la semantica operacional de Prolog. Esto
es, las sub-metas se deben resolver de izquierda a derecha, y las cláusulas se buscan
en el orden en que aparecen en el texto del programa.

En lo que sigue asumiremos que not/1 es un predicado predefinido en Prolog de-
finido como un operador prefijo. De hecho SWI-Prolog, define el operador \+ para
ello. De forma que los ejemplos anteriores, se pueden definir ahora de la siguiente
forma:

1 %% % Versiones con not

2

3 le_gusta2(adriana,X) :-

4 animal(X),

5 \+ serpiente(X).

6

7 diferente2(X,Y) :-

8 \+ (X=Y).

6.2.1 CWA, NAF y corte

Los ejemplos de la sección anterior ilustran las ventajas y desventajas de usar el
operador de corte. Las primeras incluyen:

1. Usando el operador de corte se puede mejorar la eficiencia de los programas
lógicos. La idea es decirle a Prolog explícitamente –No intentes otra alternati-
va, pues están condenadas al fracaso.

2. Usando el operador de corte podemos definir cláusulas mutuamente exclusi-
vas, por lo que podemos expresar reglas de la forma: Si la condición P enton-
ces conclusión Q, y en cualquier otro caso R. Mejorando así la expresividad
del lenguaje.

Las reservas con respecto al uso del operador de corte tienen que ver con que
fácilmente podemos perder la correspondencia entre el significado declarativo de
nuestros programas y su significado procedural. Esto es, si cambiamos el orden
de las cláusulas de nuestro programa, sin usar el operador de corte, afectamos
la eficiencia o las condiciones de terminación de éste, sin cambiar su significado
declarativo; pero si las cláusulas usan corte, un cambio en su orden si que afecta
éste significado. Veamos un ejemplo:

Ejemplo 6.7. Si queremos expresar en Prolog p⇔ (a ∧ b) ∨ c podemos usar el siguiente
programa:

1 p :- a,b.

2 p :- c.

Podemos cambiar el orden de las cláusulas y el significado declarativo del programa sigue
siendo el mismo. Introduzcamos un corte:

1 p :- a, !, b.

2 p :- c.

El significado declarativo del programa es ahora p ⇔ (a ∧ b) ∨ (¬a ∧ c). Pero si inverti-
mos el orden de las cláusulas el significado declarativo se vuele p⇔ c ∨ (a ∧ b).

Como se mencionó, los casos en que el corte cambia el significado declarativo del
programa, se denominan cortes rojos; en contraste con los cortes verdes que no lo
hacen.

Ahora bien, la definición de la negación usando fail y corte, tiene un problema
más serio: No se corresponde con la definición lógica de la negación. Consideren el
siguiente ejemplo:

Ejemplo 6.8. Consideren el siguiente programa de una sola cláusula:

1 pintor(artur_heras).

Si le preguntamos al programa si artur_heras es un pintor:

1 ?- pintor(artur_heras).

2 true

Lo cual es correcto pues esa respuesta se sigue del programa. Ahora, si preguntamos:

1 ?- pintor(juan_gris).

2 false

Lo cual también es correcto puesto que no se sigue del programa que juan_gris sea un
pintor. Ahora:

1 ?- \+ pintor(juan_gris).

2 true

En este caso, el true de Prolog no significa que la meta se siga lógicamente del programa.

La última respuesta de Prolog se basa en el supuesto del mundo cerrado (CWA). CWA

i.e., un programa Prolog representa todo lo que es verdadero en el mundo que des-
cribe. De manera que aquello que no esté declarado en el programa, o sea derivable
de éste lógicamente, se asume como falso; y su negación en consecuencia se asu-
me verdadera. El CWA puede formularse como una pseudo-regla de inferencia que
expresa:

∆ 6` α

¬α
(CWA)

Si una fbf atómica de base (sin variables) α, no puede derivarse del programa
∆ siguiendo las reglas de inferencia del sistema, entonces puede derivarse ¬α. En
el caso de los sistemas correctos y completos, la condición ∆ 6` α es equivalente
a ∆ 6|= α. Como este es el caso para la resolución-SLD, la condición puede ser
remplazada por α 6∈ M∆.

Ejemplo 6.9. Dado el siguiente programa lógico:

sobre(X, Y) ← en(X, Y).

sobre(X, Y) ← en(X, Z), sobre(Z, Y).

en(c, b).

en(b, a).

la fbf sobre(b, c) no puede ser derivada por resolución-SLD a partir del programa ∆ (vean
el árbol de derivación en la Figura 6.4). En realidad sobre(b, c) no puede ser derivada por
ningún sistema correcto, puesto que no es una consecuencia lógica de ∆. Dada la comple-
titud de la resolución-SLD, se sigue que ∆ 6|= sobre(b, c) y usando la CWA inferimos que
¬sobre(b, c).

En contra de lo que podría ser nuestra primera intuición, existen problemas aso-
ciados a la CWA. Uno de ellos es que en la vida cotidiana existen muchas situaciones
en donde el CWA no se puede asumir.

Otro problema, técnico en este caso, tiene que ver con que establecer que una meta
no es derivable dado un programa definitivo es no decidible en el caso general. Esto
es, no es posible determinar si la pseudo-regla asociada al CWA aplica o no. Una
versión más débil de la suposición de mundo cerrado, se logra si asumimos que ¬α
es derivable a partir del programa ∆ si la meta ← α tiene un árbol-SLD finito que
falla, i.e., NAF.

← sobre(b, c)

← en(b, c) ← en(b, Z0), sobre(Z0, c)

← sobre(a, c)

← en(a, c) ← en(a, Z2), sobre(Z2, c)

Figura 6.4: Árbol de derivación-SLD fallido

Es necesario contrastar la NAF con el CWA, que también puede verse como
una negación por falla, pero infinita. Para ilustrar la diferencia entre los dos enfo-
ques extendamos el programa ∆ con la siguiente cláusula evidentemente verdadera
sobre(X, Y)← sobre(X, Y).

Ejemplo 6.10. El árbol-SLD de la meta← sobre(b, c) sigue sin contener refutaciones, pero
ahora es infinito. Por lo tanto no podemos concluir que ¬sobre(b, c) usando NAF, pero si
usando CWA.

Otro aspecto a considerar es el efecto de la negación en los cuantificadores que
están bajo su alcance. Veamos un ejemplo:

Ejemplo 6.11. Consideren el siguiente programa acerca de restaurantes:

1 estrellas(diverxo).

2 estrellas(ricard_camarena).

3

4 caro(diverxo).

5

6 razonable(X) :-

7 \+ caro(X).

Una meta de interés sería:

1 ?- estrellas(X), razonable(X).

2 X = ricard_camarena.

Ahora bien, si preguntamos aparentemente la misma meta:

1 ?- razonable(X), estrellas(X).

2 false.

Usen la traza para entender este comportamiento de Prolog... y siempre prioricen calidad
sobre precio :-)

Como recordarán, la diferencia entre estas dos metas que, aparentemente equiva-
lentes, es que en el primer caso, la variable X está instanciada cuando razonable/1
es invocada; mientras que en el segundo caso esto no es así.

La cuestión es que la negación cambia la cuantificación de la variable bajo su
alcance. La meta caro(X) significa ¿Existe una X tal que caro(X) es verdadera? Si
ese es el caso ¿Cual X? La meta notcaro(X) significa ¿Verdad que para todo X, no
es el caso que caro(X)? ó ¿Es cierto que nadie es caro? En el ejemplo anterior, dado
que diverxo es caro, la respuesta es f alse. En general notMeta es seguro si todas las
variables en Meta están instanciadas.

EL objetivo de advertir esos problemas con el operador de corte, e indirectamente
con la negación, no es desalentar su uso. Por el contrario, pueden ser de gran utili-
dad. La cuestión es usarlos de manera adecuada, atendiendo sus singularidades.

Existen dos aproximaciones a la solución de estos problemas: ver los programas
como resúmenes de programas más extensos que validan las literales negativas; o

redefinir la noción de consecuencia lógica de forma que sólo algunos modelos del
programa (el mínimo de Herbrand, por ejemplo) sean tomados en cuenta. En ambos
casos, el efecto es descartar algunos modelos del programa que no son interesantes.
Primero justificaremos la regla NAF en términos de la compleción de los programas
definitivos y posteriormente, extenderemos el lenguaje de los programas definitivos
para incluir en ellos literales negativas en la cabeza y cuerpo de las cláusulas.

6.3 la compleción de un programa
La idea que presentaremos a continuación se debe a Clark [25] y se basa en que
cuando uno escribe un programa definitivo ∆, en realidad quiere expresar algo más
que su conjunto de cláusulas definitivas. El programa deseado puede formalizarse
como la compleción de ∆. Consideren la siguiente definición:

sobre(X, Y) ← en(X, Y).

sobre(X, Y) ← en(X, Z), sobre(Z, Y).

Estas reglas especifican que un objeto está sobre un segundo objeto, si el primer
objeto está encima del segundo (1) ó si el objeto está sobre otro objeto que a su vez
está encima del segundo (2). Esto también puede escribirse como:

sobre(X, Y)← en(X, Y) ∨ (en(X, Z), sobre(Z, Y))

Ahora, ¿Qué sucede si remplazamos la implicación por la equivalencia lógica?

sobre(X, Y)↔ en(X, Y) ∨ (en(X, Z), sobre(Z, Y))

Está fbf expresa que X está sobre Y si y sólo si una de las condiciones es verdadera.
Esto es, si ninguna de las condiciones se cumple, ¡se sigue que X no está sobre Y!
Esta es la intuición seguida para justificar la negación por fallo.

Desafortunadamente, combinar cláusulas definitivas como en el ejemplo anterior,
sólo es posible para cláusulas con cabezas idénticas. Por ejemplo:

en(c, b).

en(b, a).

Por una simple transformación, el programa puede ser escrito como:

en(X1, X2) ← X1 = c, X2 = b

en(X1, X2) ← X1 = b, X2 = a

Las cláusulas pueden combinarse en una sola fórmula, donde la implicación es
remplazada por la equivalencia lógica.

en(X1, X2)↔ (X1 = c, X2 = b) ∨ (X1 = b, X2 = a)

La lectura lógica de esta fbf es que X1 está en X2 si y sólo si X1 = c y X2 = b o si
X1 = b y X2 = a. Esta transformación se puede realizar sobre un programa lógico
definitivo ∆ y el resultado se conoce como compleción de ∆. Compleción

Definición 6.1 (Compleción). Sea ∆ un programa lógico definitivo. La compleción comp(∆)
de ∆ es el conjunto de fórmulas obtenido a partir de las siguientes tres transformaciones:

1. Para cada símbolo de predicado φ remplazar la cláusula α de la forma:

φ(t1, . . . , tm)← α1, . . . , αn (n ≥ 0)

por la fórmula:

φ(X1, . . . , Xm)← ∃Y1, . . . , Yi(X1 = t1, . . . , Xm = tm, α1, . . . , αn)

donde las Yi son todas variables en α y las Xi son variables únicas que no aparecen en
α.

2. Para cada símbolo de predicado φ remplazar todas las fbf:

φ(X1, . . . , Xm)← β1

...

φ(X1, . . . , Xm)← β j

por la fórmula:

∀X1, . . . , Xm(φ(X1, . . . , Xm)↔ β1∨, . . . ,∨β j si j > 0

∀X1, . . . , Xm(¬φ(X1, . . . , Xm)) si j = 0

3. Finalmente el programa se extiende con los siguientes axiomas de igualdad libre, que
definen las igualdades introducidas en el paso 1:

∀(X = X)

∀(X = Y ⇒ Y = X)

∀(X = Y ∧Y = Z ⇒ X = Z)

∀(X1 = Y1 ∧ · · · ∧ Xn = Yn ⇒ f (X1, . . . , Xn) = f (Y1, . . . , Yn))

∀(X1 = Y1 ∧ · · · ∧ Xn = Yn ⇒ (φ(X1, . . . , Xn)⇒ φ(Y1, . . . , Yn))

∀( f (X1, . . . , Xn) = f (Y1, . . . , Yn)⇒ X1 = Y1 ∧ · · · ∧ Xn = Yn)

∀(¬ f (X1, . . . , Xm) = g(Y1, . . . , Yn))(Si f /m 6= g/n)

∀(¬X = t)(Si X es un subtermino propio de t)

Estas definiciones garantizan que la igualdad (=) sea una relación de equivalen-
cia; que sea una relación congruente; y que formalice la noción de unificación. Las
primeros cinco definiciones se pueden abandonar si se especifica que = representa
la relación de identidad. Identidad

Ejemplo 6.12. Consideremos la construcción de comp(∆) tal y como se definió anterior-
mente. El primer paso produce:

sobre(X1, X2) ← ∃X, Y (X1 = X, X2 = Y, en(X, Y))

sobre(X1, X2) ← ∃X, Y, Z (X1 = X, X2 = Y, en(Z, Y), sobre(Z, Y))

en(X1, X2) ← (X1 = c, X2 = b)

en(X1, X2) ← (X1 = b, X2 = a)

dos pasos más adelante obtenemos:

∀X1, X2(sobre(X1, X2 ↔ ∃X, Y(. . . ) ∧ ∃X, Y, Z(. . . ))

∀X1, X2(en(X1, X2) ↔ (X1 = c, X2 = b) ∧ (X1 = b, X1 = a))

y el programa se termina con las definiciones de igualdad como identidad y unificación.

La compleción comp(∆) de un programa definitivo ∆ preserva todas las literales
positivas modeladas por ∆. Esto es, si ∆ |= α entonces comp(∆) |= α. Tampoco se
agrega información positiva al completar el programa: Si comp(∆) |= α entonces
∆ |= α. Por lo tanto, al completar el programa no agregamos información positiva
al mismo, solo información negativa.

Como sabemos, no es posible que una literal negativa pueda ser consecuencia
lógica de un programa definitivo. Pero al substituir las implicaciones en ∆ por equi-
valencias en comp(∆) es posible inferir información negativa a partir del programa
completado. Esta es la justificación de la regla NAF, cuyas propiedades de consis-
tencia se deben a Clark [25]:

Teorema 6.1 (Consistencia de la NAF). Sea ∆ un programa definitivo y ← α una meta
definitiva. Si← α tiene un árbol-SLD finito fallido, entonces comp(∆) |= ∀(¬α).

La consistencia se preserva aún si α no es de base. Por ejemplo,← en(a, X) falla de
manera finita y por lo tanto, se sigue que comp(∆) |= ∀(¬en(a, X)). La completitud
de la NAF también ha sido demostrada:

Teorema 6.2 (Completitud de la NAF). Sea ∆ un programa definitivo. Si comp(∆) |=
∀(¬α) entonces existe un árbol finito fallido para la meta definitiva← α.

Observen que solo enuncia la existencia de un árbol-SLD finito fallido. Como se
ha mencionado, un árbol-SLD puede ser finito bajo ciertas reglas de computación
e infinito bajo otras. En particular, el teorema de completitud no es válido para
las reglas de computación de Prolog. La completitud funciona para una subclase
de derivaciones-SLD conocidas como justas (fair), las cuales o bien son finitas o Derivación justa

garantizan que cada átomo en la derivación (u ocurrencia de éste), es seleccionado
eventualmente por las reglas de computación. Un árbol-SLD es justo si todas sus
derivaciones son justas. La NAF es completa para árboles-SLD justos. Este tipo de
derivaciones se pueden implementar fácilmente: selecciona la sub-meta más a la
izquierda y agrega nuevas submetas al final de esta (búsqueda en amplitud). Sin
embargo, pocos sistemas implementan tal estrategia por razones de eficiencia.

6.4 resolución sldnf para programas definitivos
En el capítulo 5.4 presentamos el método de resolución-SLD, utilizado para probar
si una literal positiva cerrada es consencuencia lógica de un programa. En la sección
anterior afirmamos que también las literales negadas pueden derivarse a partir de la
terminación de programas lógicos definitivos. Combinando la resolución SLD y la
negación como fallo finito (NAF), es posible generalizar la noción de meta definitiva
para incluir literales positivas y negadas. Tales metas se conocen como generales.

Definición 6.2 (Meta general). Una meta general tiene la forma:

← α1, . . . αn (n ≥ 0)

donde cada αi es una literal positiva o negada.

La combinación de la resolución SLD y la NAF se llama resolución SLDNF.

Definición 6.3 (Resolución SLDNF para programas definitivos). Sea ∆ un programa
definitivo, G0 una meta general y R una función de selección (también conocida como regla
de computación). Una derivación SLDNF de G0 usandoR, es una secuencia finita o infinita
de metas generales:

G0
α0 G1 . . . Gn−1

αn−1 Gn

donde Gi
αi Gi+1 puede ocurrir si:

1. la literal R-seleccionada en Gi es positiva y Gi+1 se deriva de Gi y αi por un paso de
resolución SLD;

2. la literal R-seleccionada en Gi es negativa (¬α) y la meta ← α tiene un árbol SLD
fallido y finito y Gi+1 se obtiene a partir de Gi eliminando ¬α (en cuyo caso αi,
corresponde al marcador especial FF).

Cada paso en una derivación SLDNF produce una substitución, en el caso 1 un
MGU y en el caso 2, la substitución vacía ε.

Entonces, una literal negativa ¬α es demostrada si← α tiene un árbol SLD finito
que falla. Por dualidad, ¬α falla de manera finita si α es demostrada. Además de
la refutación y de la derivación infinita, existen dos clases de derivaciones SLDNF
completas dada una función de selección:

1. Una derivación se dice (finitamente) fallida si (i) la literal seleccionada es
positiva y no unifica con ninguna cabeza de las cláusulas del programa, o
(2) la literal seleccionada es negativa y tiene un fallo finito.

2. Una derivación se dice plantada (stuck) si la sub-meta seleccionada es de la
forma ¬α y← α tiene un fallo infinito.

Ejemplo 6.13. Considere el siguiente programa:

en(c, b)

en(b, a)

La meta ← en(X, Y),¬en(Z, X) tiene una refutación-SLDNF con la substitución compu-
tada {X/c,Y/b}:

G = ← en(X, Y),¬en(Z, X).

G0 = ← en(X, Y).

α0 = en(c, b).

θ0 = {X/c, Y/b}

G1 = ¬en(Z, X)θ0 =← en(Z, c)

α1 = FF

θ1 = ε

G2 = �

θ = θ0θ1 = {X/c, Y/b}

En cambio, si la función de selección hubiera computado las cláusulas de abajo hacía arriba
α0 = en(b, a) la derivación hubiera sido fallida (a ustedes probarlo).

Como es de esperarse, la resolución-SLDNF es consistente, después de todo, la
resolución-SLD y la NAF son consistentes.

Teorema 6.3 (Consistencia de la resolución-SLDNF). Sea ∆ un programa definitivo y
← α1, . . . , αn una meta general. Si ← α1, . . . , αn tiene una refutación SLDNF con una
substitución computada θ, comp(∆) |= ∀(α1θ, . . . , αnθ).

Sin embargo, la resolución-SLDNF no es completa aunque pudiéramos haber
esperado lo contrario. La resolución SLDNF no es completa a pesar de que la
resolución-SLD y la NAF si lo son. Un simple contra ejemplo es ← ¬en(X, Y) que
corresponde a la consulta “¿Hay algunos bloques X e Y, tal que X no está en Y?”

Uno esperaría varias respuestas a esta consulta, por ejemplo, el bloque a no está
encima de ningún bloque, etc.

Pero la derivación SLDNF de← ¬en(X, Y) falla porque la meta← en(X, Y) tiene
éxito (puede ser demostrada). El problema es que nuestra definición de derivación
fallida es demasiado conservadora. El éxito de ← en(X, Y) no significa necesaria-
mente que no halla un bloque que no esté en otro bloque, sólo que existe al menos
un bloque que no está en otro.

El problema tiene su origen en que la NAF, en contraste con la resolución SLD,
es sólo una prueba (test). Recuerden que dada la definición de la resolución SLDNF
y la consistencia y completitud de la NAF, tenemos que ¬en(X, Y) tiene éxito si
y sólo si (≡) en(X, Y) tiene asociado un árbol SLD fallido y finito; o si y sólo si
comp(∆) |= ∀(¬en(X, Y)). Por lo tanto, la meta general ← en(X, Y) no debe leerse
como una consulta cuantificada existencialmente, sino como una prueba universal:
“Para todo bloque X e Y, ¿No está X en Y?”.

Esta última consulta tiene una respuesta negativa en el modelo deseado del pro-
grama, puesto que el bloque b está en el bloque a. El problema anterior se debe a
la cuantificación de las variables en la literal negativa. Si replanteamos la consulta
anterior como ← ¬en(a, b) entonces la resolución SLDNF alcanza una refutación
puesto que← en(a, b) falla con una derivación finita.

Algunas veces se asume que la función de selección R permite seleccionar una
literal negativa ¬α si la literal α no tiene variables libres o si α tiene asociada una
substitución computada vacía. Estas funciones de selección se conocen como segu-
ras (safe).

6.5 programas lógicos generales
Con los desarrollos anteriores estamos en posición de extender el lenguaje de los
programas definitivos para incluir cláusulas que contienen literales tanto positivas
como negativas en su cuerpo. Estas fbf se llaman cláusulas generales, y en conse- Cláusulas Generales

cuencia, un conjunto de ellas conforma un programa general. En ocasiones, a los
programas generales se les conoce a veces como programas normales. Programas Normales

Definición 6.4 (Cláusula General). Una cláusula general es una fbf de la forma α ←
α1, . . . , αn donde α es una fbf atómica y α1, . . . , αn son literales (n ≥ 0).

Definición 6.5 (Programa General). Un programa lógico general es un conjunto finito
de cláusulas generales.

Ahora podemos extender nuestro programa del mundo de los bloques con las
siguientes relaciones:

base(X) ← en(Y, X), en_la_mesa(X).

en_la_mesa(X) ← ¬no_en_la_mesa(X).

no_en_la_mesa(X) ← en(X, Y).

en(c, b).

en(b, a).

La primer cláusula especifica que un bloque es base si está sobre la mesa y tiene
otro bloque encima. La segunda cláusula indica que cuando no es cierto que un blo-
que no está sobre la mesa, entonces está sobre la mesa. La tercera cláusula especifica
que un bloque que está sobre otro, no está sobre la mesa.

Parece claro, pero la pregunta que deberíamos hacernos es qué tipo de sistema
de prueba queremos para los programas lógicos generales y cuales serán las apro-
ximaciones lógicas a las sutilezas, algunas ya discutidas, introducidas por este tipo
de lenguajes.

Observen que aunque el lenguaje fue enriquecido, no es posible de cualquier
forma que una literal negativa sea consecuencia lógica de un programa dado. La
razón es la misma que para los programas definidos, la base de Herbrand de un
programa ∆, B∆ es un modelo de ∆ en el que todas las literales negativas son falsas.
Al igual que con los programas definidos, la pregunta es entonces como lograr
inferencias negativas consistentes. Afortunadamente el concepto de compleción de
programa puede aplicarse también a los programas lógicos generales.

Ejemplo 6.14. La compleción de gana(X)← mueve(X, Y),¬gana(Y) contiene la fbf:

∀X1(gana(X1)↔ ∃X, Y(X1 = X, mueve(X, Y),¬gana(Y)))

Desafortunadamente, la compleción de los programas normales puede ocasionar
paradojas.

Ejemplo 6.15. Consideren la cláusula general p ← ¬p, su compleción incluye p ↔ ¬p.
La inconsistencia del programa terminado se debe a que p/0 está definida en términos de su
propio complemento.

Una estrategia de programación para evitar este problema consiste en componer
los programas por capas o estratos, forzando al programador a referirse a las nega-
ciones de una relación hasta que ésta ha sido totalmente definida. Se entiende que
tal definición se da en un estrato inferior a donde se presenta la negación. En la
definición del programa estratificado usaremos ∆p para referirnos al subconjunto Programa

Estratificadode cláusulas en ∆ que tienen a p como cabeza.

Definición 6.6 (Programa Estratificado). Un programa general ∆ se dice estratificado si
y sólo si existe al menos una partición ∆1 ∪ · · · ∪ ∆n de ∆ tal que :

1. Si p(. . . )← q(. . . ), · · · ∈ ∆i entonces ∆q ⊆ ∆1 ∪ · · · ∪ ∆i;

2. Si p(. . . )← ¬q(. . . ), · · · ∈ ∆i entonces ∆q ⊆ ∆1 ∪ · · · ∪ ∆i−1.

Por ejemplo, el siguiente programa está estratificado:

∆2: base(X)← en(Y, X), en_la_mesa(X).
en_la_mesa(X)← ¬no_en_la_mesa(X).

∆1: no_en_la_mesa(X)← en(X, Y).
en(c, b).
en(b, a).

Apt, Blair y Walker [2] demostraron que la compleción de un programa estrati-
ficado en consistente, de forma que la situación descrita anteriormente no puede
ocurrir. Sin embargo, la estratificación es solo una condición suficiente para la con- Consistencia

sistencia: Determinar si un programa está estratificado es decidible, pero determi-
nar si la compleción de un programa general es consistente, es indecidible. Por lo
tanto, hay programas generales no estratificados cuya compleción es consistente.

6.6 resolución-sldnf para programas generales
Hemos revisado el caso de la resolución-SLDNF entre programas definitivos y me-
tas generales. Informalmente podemos decir que la resolución-SLDNF combina la
resolución-SLD con los siguientes principios:

1. ¬α tiene éxito ssi← α tiene un árbol-SLD finito que falla.

2. ¬α falla finitamente ssi← α tiene una refutación-SLD.

El paso de programas definitivos a programas generales, es más complicado. Pa-
ra probar ¬α, debe de existir un árbol finito fallido para ← α. Tal árbol puede
contener nuevas literales negativas, las cuales a su vez deben pueden tener éxito o
fallar finitamente. Esto complica un poco la definición de la resolución-SLDNF para
programas generales.

Ejemplo 6.16. Es posible llegar a situaciones paradójicas cuando los predicados están defi-
nidos en términos de sus propios complementos. Consideren el programa no estratificado:

α← ¬α

Dada la meta inicial ← α, se puede construir una derivación ← α  ← ¬α. La deriva-
ción puede extenderse hasta una refutación si← α falla finitamente. De manera alternativa,
si ← α tiene una refutación, entonces la derivación falla. Helas! esto es imposible pues la
meta← α no puede tener una refutación y fallar finitamente al mismo tiempo.

En lo que sigue, definiremos las nociones de derivación-SLDNF y árbol-SLDNF,
de manera similar a la derivación-SLD y a los arboles-SLD. La idea se concreta en el
concepto de bosque-SLDNF: Un conjunto de árboles cuyos nodos está etiquetados Bosque-SLDNF

con metas generales. Un sub-bosque se obtiene removiendo algunos nodos y sus Sub-Bosque
hijos del bosque original. Dos sub-bosques B1 y B2 se consideran equivalentes si
contieinen los mismos árboles considerando un posible renombrado de variables.
B1 es más pequeño que B2 si es equivalente a algún sub-bosque de B2.

Definición 6.7 (Bosque-SLDNF). Sea ∆ un programa general, G0 una meta general, y R
una función de selección. El bosque-SLDNF de G0 es el bosque más pequeño (considerando
el posible renombrado de variables), tal que:

1. G0 es la raíz del árbol.

2. Si G es un nodo en el bosque cuya literal seleccionada es positiva, entonces para cada
cláusula α tal que G′ puede ser derivada de G y α (con UMG θ), G tiene un hijo
etiquetado como G′. Si no existe tal cláusula, entonces G tiene un hijo etiquetado FF
(falla finita);

3. Si G es un nodo del bosque cuya literal seleccionada es de la forma ¬α (G es de la
forma← α1, . . . , αi−1,¬α, αi+1, . . . , αn), entonces:

• El bosque contiene un árbol cuya raíz está etiquetada como← α;

• Si el árbol con raíz ← α tiene una hoja � con la substitución computada ε,
entonces G tiene un sólo hijo etiquetado FF;

• Si el árbol con raíz ← α es finito y tiene todas sus hojas etiquetadas FF, en-
tonces G tiene un sólo hijo (con substitución asociada ε) etiquetado como ←
α1, . . . , αi−1, αi+1, . . . , αn.

Observen que la literal negativa seleccionada ¬α falla sólo si← α tiene una refuta-
ción con la substitución computada ε. Como veremos más adelante, esta condición
que no era necesaria cuando definimos la resolución-SLDNF para programas defi-
nitivos, es vital para la correctez de la resolución en los programas generales. Correctez

Los arboles del bosque-SLDNF son llamados arboles-SLDNF (completos); y la Arboles-SLDNF

secuencia de todas las metas en una rama de un árbol-SLDNF con raíz G es llamada
derivación-SLDNF completa de G (bajo un programa ∆ y una función de selección
R). El árbol etiquetado por G0 es llamado árbol principal. Un árbol con la raíz← α
es llamado árbol subsidiario si ¬α es una literal seleccionada en el bosque (el árbol
principal puede ser a su vez subsidiario).

Ejemplo 6.17. Consideren el siguiente programa general estratificado ∆:

base(X) ← en(Y, X), en_la_mesa(X).

en_la_mesa(X) ← ¬no_en_la_mesa(X).

no_en_la_mesa(X) ← en(X, Y).

encima(X, Y) ← en(X, Y).

encima(X, Y) ← en(X, Z), encima(Z, Y).

en(c, b).

en(b, a).

El bosque-SLDNF para la meta← base(X) se muestra en la Figura 6.5. El árbol principal
contiene una derivación fallida y una refutación con la substitución computada {X/a}.

base(X).

← en(Y0, X), en_la_mesa(X).

← en_la_mesa(b).

← ¬no_en_la_mesa(b).

FF

← en_la_mesa(a).

← ¬en_la_mesa(a).

�

← no_en_la_mesa(b).

← en(b, Y0).

�

← no_en_la_mesa(a).

← en(a, Y0).

FF

Figura 6.5: Bosque-SLDNF para la meta← base(X).

Las ramas de un árbol-SLDNF en un bosque-SLDNF representan todas las derivaciones-
SLDNF completas de su raíz, con base en la función de selección dada. Hay cuatro
clases de derivaciones-SLDNF completas:

1. Derivaciones infinitas;

2. Derivaciones finitas fallidas (terminan en FF);

3. Refutaciones (terminan en �); y

4. Derivaciones plantadas (si ninguno de los casos anteriores aplica).

Ejemplo 6.18. Consideren el siguiente programa:

termina(X) ← ¬ciclo(X).

ciclo(X) ← ciclo(X).

El bosque-SLDNF para el ejemplo anterior se muestra en la Figura 6.6. El bosque
incluye una derivación plantada para termina(X) y una derivación infinita para

← termina(X).

← ¬ciclo(X).

← ciclo(X).

← ciclo(X).

∞

Figura 6.6: Bosque-SLDNF para la meta← paro(X).

ciclo(X). Esto ilustra una de las razones por las cuales una derivación se planta:
uno de sus arboles subsidiarios contiene sólo derivaciones fallidas o infinitas.

El siguiente programa también conduce a una derivación plantada (ciclo en el
cómputo de la negación):

paradoja(X) ← ¬ok(X).

ok(X) ← ¬paradoja(X).

Intenten construir el bosque-SLDNF de este programa y observaran también que
en este caso, la árbol principal es a su vez un árbol subsidiario.

La última razón para que una derivación quede plantada es ilustrada por el si-
guiente programa:

tope(X) ← ¬bloqueado(X).

bloqueado(X) ← en(Y, X).

en(a, b).

Es evidente que tope(a) debería poder derivarse de este programa. Sin embargo,
el árbol-SLDNF de la meta← tope(X) no contiene refutaciones. De hecho, esta me-
ta se planta aún cuando ← bloqueado(X) tiene una refutación. La razón de esto es
que← bloqueado(X) no tiene ninguna derivación que termine con una substitución
computada vacía. A la meta ← ¬tope(X), Prolog no responde b, sino que ¡no to-
dos los bloques están en el tope de la pila! De manera que la implementación de
la resolución-SLDNF en la mayoría de los Prolog no es robusta, aunque nuestra
definición si lo es.

Teorema 6.4 (Correctez de la resolución-SLDNF). Sea ∆ un programa general y ←
α1, . . . , αn una meta general. Entonces:

• Si← α1, . . . , αn tiene una substitución de respuesta computada θ, entonces comp(∆) |=
∀(α1θ ∧ · · · ∧ αnθ).

• Si← α1, . . . , αn tiene un árbol-SLDNF finito que falla, entonces comp(∆) |= ∀(¬(α1∧
· · · ∧ αn)).

La definición de bosque-SLDNF no debe verse como una implementación de la
resolución-SLDNF, sólo representa el espacio ideal de computación donde la correc-
tez puede ser garantizada. Nada se ha dicho en cuanto al orden en que el bosque
debe construirse. Al igual que en los demás casos, Prolog sigue una estrategia pri-
mero en profundidad: Ante una meta ¬α, Prolog suspende la construcción del árbol
de resolución en espera de la respuesta a la meta ← α. El hecho de que Prolog no
verifique si la substitución de respuesta de una refutación fue ε, al igual que la
ausencia de chequeo de ocurrencias, contribuyen a que Prolog no sea robusto.

6.7 lecturas y ejercicios sugeridos
La discusión presentada aquí sobre corte y negación por fallo finito en Prolog, se
basa en el libro de Bratko [19]. Los aspectos generales más técnicos sobre estos
temas se basan en el libro de Nilsson y Maluszynski [88]. Apt y Bol [3] nos ofrecen
una revisión de la negación en el contexto de la programación lógica.


